
1. O que este documento é

Vinte e duas deducdes tensoriais de A First Course in General Relativity (B. F. Schutz, 32
ed., Cambridge, 2022), refeitas pelo ODEROM. Cada uma traz a secdo e a pagina do livro,
o documento .od completo, o comando, e a saida do programa — bytes reais, capturados
do bindrio, nao transcritos.

As contas foram escolhidas por serem afirmacoes verificaveis: o livro diz quanto vale
alguma coisa, e o programa ou reproduz ou nao. Duas delas sao do tipo "one can show
that", em que o autor pula a dlgebra — e sdo justamente as que mais valem a pena.

Elas vém em duas partes, e a divisao nao é cronoldgica. As doze dos capitulos 2 a 5 sao do
tipo "calcule este tensor e compare com a féormula": o livro dd uma expressao
fechada e o programa tem de chegar nela. As dez da parte 7 sao de outra natureza —
identidades e fatos estruturais: simetrias do Riemann, as identidades de Bianchi,
curvatura nula onde a conexao nao é nula, matéria em espaco plano, e uma solucdo exata
gue nao é esfericamente simétrica.

Como cada conta é conferida

Onde o livro da uma férmula fechada, ela é escrita como um tensor no préprio documento
e subtraida do que o ODEROM calcula. Os dois lados passam pelo mesmo parser e pelo
mesmo normalizador, e o que se cobra é que a diferenca seja identicamente nula.

Comparar strings de saida provaria bem menos: passaria a depender de escolhas de
renderizacao — se 0 programa escreve 1/r ou r~-1, se pde a exponencial no numerador
ou no denominador — que nao sao o assunto. A diferenca nao depende de nada disso.

Onde o zero é fraco, esta dito. Uma identidade que vale por construcao nao prova que o
programa a calculou: se V devolvesse zero para tudo, a divergéncia do Einstein seria nula de
graca. Por isso as contas desse tipo vém com um controle ao lado — uma quantidade que ndo
é nula e cujo valor o livro também da. Sem o controle, o zero nao é evidéncia.

Como rodar o que esta aqui

Cada conta aparece em dois blocos, que é como o caderno funciona: um com as
declaracdes, e o seguinte com a pergunta. Copie o primeiro, Sshift + Enter , copie o
segundo, shift + Enter . Onde ha mais de uma pergunta, cada linha é um bloco.



Na linha de comando é outra grafia, e ndo é a que este documento mostra. O caderno
pede eval V[a;a] ; o terminal pede o arquivo e as aspas do shell:

oderom eval --prelude polar.od "V[a;a]"

Uma consulta sem expressao — christoffel, riemann, einstein — vira oderom
christoffel polar.od . Colar a linha do terminal num bloco do caderno nao funciona: oderom
nao é palavra da linguagem, e a recusa lista as que sao.

Todo documento .od deste relatério estd completo no texto. As vinte e duas contas
também estdo fixadas como testes em oderom-cli/tests/schutz.rs, com as mesmas
referéncias de pagina, e rodam em cerca de sete segundos com cargo test -p oderom-cli
--test schutz.

2. Capitulo 5 — calculo tensorial em coordenadas
polares

O plano euclidiano numa carta curvilinea: sem curvatura, mas com simbolos de Christoffel ndo nulos. E
onde o livro separa "a conexao nao é zero" de "o espaco é curvo".

2.1 Os simbolos de Christoffel em coordenadas polares | conrere
Schutz: §5.4, p.127, Eq. 5.45
No plano euclidiano em coordenadas polares, trés dos oito simbolos séo n&do nulos: Mgg =

-re Fere = Feer = 1/r. Schutz os obtém derivando os vetores de base (de, /90 = eg/r, 0eg/06

= —r e,); aqui saem da métrica pela féormula usual.

O documento — um bloco.

manifold M dim 2
bundle TM on M dim 2

chart pol on M coords (r, theta)
metric g on pol bundle TM { [r,r] = 1, [theta,theta]l = r*2 }

A pergunta — o bloco seguinte, e a resposta.

christoffel

Gamma[”~r, theta, thetal
Gamma[~theta, r,thetal r~-1
Gamma[~theta, theta, r] r~-1
5 independent components identically zero

-r

Os trés do livro, com 0os mesmos sinais, e a confirmacado de que os outros cinco se anulam
— que é metade da afirmacao e a metade que costuma ficar sem conferir.



2.2 A divergéncia de um vetor em polares | conFeRe
Schutz: §5.3, p.129, Eq. 5.56

V%o = oVijor + aV9/30 + V'/r. O terceiro termo é o que a derivada parcial ndo d&, e vem de

%, = 1/r.

O documento — um bloco.

manifold M dim 2
bundle TM on M dim 2

chart pol on M coords (r, theta)
metric g on pol bundle TM { [r,r] = 1, [theta,theta]l = r*2 }
tensor V : TM on pol { [r] = A(r,theta), [theta] = B(r,theta) }

A pergunta — o bloco seguinte, e a resposta.

eval V[a;al

[ (r*A r(r,theta) + r*B theta(r,theta) + A(r,theta))/r

Dividindo por r: A r + B theta + A/r, que é a Eq. 5.56. As componentes sao funcdes
indeterminadas das duas coordenadas de propdsito — com um vetor concreto a conta
fecharia por acidente aritmético em vez de reproduzir a férmula.

2.3 O laplaciano em polares | CONFERE
Schutz: §5.3, p.129, Eq. 5.57

V2@ = (1/r) o/(r 9,9) + (1/r?) a2gep. O caminho do livro é: converter a 1-forma gradiente em
vetor, o que da componentes (¢ [, ¢ g/r?), e aplicar a Eq. 5.56.

O documento — um bloco.

manifold M dim 2
bundle TM on M dim 2

chart pol on M coords (r, theta)
metric g on pol bundle TM { [r,r] = 1, [theta,theta] = r*2 }
tensor D : TM on pol { [r] = phi {r}(r,theta), [theta] = phi {theta}(r,theta)/r"2 }

A pergunta — o bloco seguinte, e a resposta.

eval D[a;al

[ (r*phi r(r,theta) + r*2*phi_rr(r,theta) + phi_thetatheta(r,theta))/r"2

Dividindo por r~2: phi_rr + phi _r/r + phi_thetatheta/r"2, que é a Eq. 5.57 com o
primeiro termo expandido. O phi {r}(r,theta) do documento é a notacao de subscrito de



derivada do ODEROM — a derivada de phi em relagao ao seu primeiro argumento,
tratada como simbolo opaco, que é exatamente o estatuto que ela tem aqui.

3. Capitulo 6 — variedades curvas

A curvatura propriamente dita: o Riemann, o que ele significa (o comutador de derivadas covariantes), e a
identidade que sustenta as equacdes de campo.

3.1 O Riemann e a curvatura da esfera CONFERE
Schutz: Exercicios 6.28-6.29, p.167

Na esfera de raio a, a unica componente independente € Rgyg, = a2 Sin%6, e 0 escalar de

Ricci é 2/a? — duas vezes a curvatura gaussiana 1/a%, como tem de ser em duas
dimensoes.

O documento — um bloco.

manifold M dim 2
bundle TM on M dim 2

chart s on M coords (theta, phi)
metric g on s bundle TM { [theta,theta] = a”2, [phi,phi] = a”2*sin(theta)”2 }
tensor Ric := ricci g

A pergunta — o bloco seguinte, e a resposta.

riemann

eval Ricl[a,a]

R[theta,phi,theta,phi] = a"2*sin(theta)”"2
(4 components by symmetry)

2/a"2

O escalar sai do traco, Ric[a,a] = gMR,,, que é a definicdo do préprio livro (Eq. 6.92).

v
Pedir tensor Rs := scalar g é recusado pelo programa, e com razao: scalar devolve um
escalar, e tensor declara tensor.

3.2 O comutador de derivadas covariantes E o Riemann CONFERE
Schutz: §6.5, p.160, Eq. 6.77

[Va, Vgl VM = RH,;g V¥. Descendo o indice livre — o que € legitimo porque Vg = 0 faz subir
e descer comutar com derivar, o argumento do préprio Schutz logo abaixo da Eq. 6.78 — a
igualdade vira [V, Vgl W, = Ry, V.

O documento — um bloco.



manifold M dim 2

bundle TM on M dim 2

chart s on M coords (theta, phi)

metric g on s bundle TM { [theta,theta] = a”2, [phi,phi] = a”2*sin(theta)”~2 }

tensor W : TM* on s { [theta] = P(theta,phi), [phi] = Q(theta,phi) }
tensor V. : TM on s { [theta] = P(theta,phi)/a”2, [phi] =
Q(theta,phi)/(a”2*sin(theta)”~2) }

tensor R := riemann g

A pergunta — o bloco seguinte, e a resposta.

eval W[m;b,a] - W[m;a,b]

eval W[m;b,al] - W[m;a,b]l - R[m,n,a,b] V[n]

[theta, theta, phi] -Q(theta, phi)
[theta,phi, thetal] Q(theta, phi)

[phi, theta,phi] = sin(theta)”2*P(theta,phi)
[phi,phi,theta] = -sin(theta)”2*P(theta,phi)
4 independent components identically zero

8 independent components identically zero

A primeira saida mostra que o comutador tem conteldo — quatro das oito células sao nao
nulas —, e a segunda que ele é exatamente o Riemann contraido. Sem a primeira, a
segunda nao provaria nada: uma segunda derivada que devolvesse zero para tudo
passaria igual.

O campo tem componentes indeterminadas P, Q, entdao nenhuma célula fecha por
acidente. Esta é a conta que exercita a segunda derivada covariante, W[m;b,a] , com 0s
dois indices de derivada em ordens trocadas.

3.3 A divergéncia do tensor de Einstein é nula | conrere
Schutz: §6.6, p.163, Eq. 6.99

G“B;B = 0, a identidade de Bianchi duas vezes contraida. E ela que permite escrever G =
8nT: o lado esquerdo é livre de divergéncia por identidade, e o direito por conservacao.

O documento — um bloco.



manifold M dim 4
bundle TM on M dim 4
chart frw on M coords (t, r, theta, phi)
metric g on frw bundle TM {
[t,t] = -1,
[r,r] = Scale(t)"2/(1 - k*r*2),

[theta,theta] = Scale(t)”"2*r"2,
[phi,phi] = Scale(t)”2*r"2*sin(theta)”"2
}
tensor G := einstein g
tensor Ric := ricci g

A pergunta — o bloco seqguinte, e a resposta.

eval G[a,b;b]

eval Ric[a,b;b]

4 independent components identically zero

[t] = (-6*k*Scale'(t) + 3*Scale(t)*Scale'(t)*Scale''(t) + 3*Scale(t)”2*Scale'''(t) -
6*Scale' (t)"3)/Scale(t)"3
3 independent components identically zero

O controle esta na segunda saida, e é o que faz a primeira valer alguma coisa. A
divergéncia do Ricci NAO é nula — o préprio livro diz quanto ela vale, 2R 5 (EQ. 6.96). Se o

programa devolvesse zero para as duas, 0 zero da primeira nao seria prova de nada.

A métrica é FRW e nao Schwarzschild justamente por isso: em vacuo G ja é nulo
componente a componente, e a divergéncia de zero é zero sem que V tenha feito trabalho
nenhum.

4. Capitulo 10 — solucoes esféricas para estrelas

A conta central de todo curso de relatividade: o tensor de Einstein da métrica estatica esfericamente
simétrica, e o que ele da quando igualado a matéria.

4.1 O tensor de Einstein da métrica estatica esfericamente simétrica
CONFERE
Schutz: §10.2, p.271, Egs. 10.14-10.17

Para ds? = —e2®dt2 + e2\dr2 + r2dQ?, o livro diz "one can show that" e escreve as quatro
componentes:

« Goo = (1/r2) e2® d/dr [r(1 — e™2N)]

Gy = —(1/r2) N1 — e72N + 2/’
« Ggg=r2e ?No" + (¥)2 + &/r — &N — AJr]
. G

ol Sin20 - Gee



Esta é a conta principal do capitulo, e o "one can show" é o que este documento existe
para mostrar.

O documento — um bloco.

manifold M dim 4
bundle TM on M dim 4
chart sph on M coords (t, r, theta, phi)
metric g on sph bundle TM {
[t,t] = -exp(2*Phi(r)),
[r,r] = exp(2*Lambda(r)),
[theta, theta] = r*2,
[phi,phi] = r*2 * sin(theta)”2
ks
tensor G := einstein g
tensor S : TM*, TM* on sph {
[t,t] exp(2*Phi(r))/r~2 * diff(r*(1 - exp(-2*Lambda(r))), r),
[r,r] -exp(2*Lambda(r))*(1 - exp(-2*Lambda(r)))/r~2 + 2*Phi'(r)/r,
[theta, theta] = r"2*exp(-2*Lambda(r))*(Phi''(r) + Phi'(r)”2 + Phi'(r)/r -
Phi' (r)*Lambda'(r) - Lambda'(r)/r),
[phi,phi] = sin(theta)”2 * r"2*exp(-2*Lambda(r))*(Phi''(r) + Phi'(r)”2 + Phi'(r)/r
- Phi'(r)*Lambda'(r) - Lambda'(r)/r)
b

A pergunta — o bloco seguinte, e a resposta.

einstein

eval G[a,b] - S[a,b]

G[t,t] = (2*r*exp(2*Phi(r))*Lambda'(r) + exp(2*Lambda(r))*exp(2*Phi(r)) -
exp(2*Phi(r)))/(r*2*exp(2*Lambda(r)))

G[r,r] = (1 + 2*r*Phi'(r) - exp(2*Lambda(r)))/r"2

G[theta,thetal = (-r*Lambda'(r) + r*Phi'(r) - r”2*Lambda'(r)*Phi'(r) + r*2*Phi'(r)”"2
+ r"2*Phi''(r))/exp(2*Lambda(r))

G[phi,phi] = (-r*sin(theta)”2*Lambda'(r) +
r~2*sin(theta)”2*Phi''(r))/exp(2*Lambda(r))
6 independent components identically zero

16 independent components identically zero

Dezesseis de dezesseis nulas. O tensor S é o do livro escrito literalmente, incluindo a
derivada d/dr[r(1 - e~{-2A})] da EqQ. 10.14, que fica a cargo de diff — nada foi
expandido a mao de um lado para bater com o outro. A conferéncia é a diferenca dos dois
tensores, e ndo a comparacao das strings de saida: assim ela ndo depende de escolhas de
renderizacao, que nao sao o assunto.

Esta conta encontrou um defeito. Escrever a Eq. 10.14 exigia diff , que o manual
documenta e que a recusa de sin'(x) manda usar — e que, até esta rodada, virava um
simbolo opaco chamado diff em vez de derivar. Ver §6.1.



4.2 A geometria exterior: Schwarzschild € vacuo | conrere
Schutz: §10.4, p.273

Fora da estrela, p = p = 0, e a solucao tem de satisfazer Ggp = 0.

O documento — um bloco.

manifold M dim 4
bundle TM on M dim 4
chart schw on M coords (t, r, theta, phi)
metric g on schw bundle TM {
[t,t] = -(1 - 2*Mass/r),

[r,r] = 1/(1 - 2*Mass/r),

[theta, theta] = r*2,

[phi,phi] = r*2 * sin(theta)”2
}

A pergunta — o bloco seguinte, e a resposta.

einstein

[ 10 independent components identically zero

As dez independentes, todas nulas. E a conta mais barata do documento e a que serve de
sanidade para todas as outras: se esta falhasse, nenhuma das demais valeria discussao.

4.3 A equacao de equilibrio hidrostatico relativistico | conrFere
Schutz: §10.3, p.272, Eqgs. 10.26-10.27

De TO‘B;B = 0, com o fluido em repouso na métrica estatica, sai (p + p) d®/dr = —dp/dr. O
livro observa que, das quatro equacdes, s6 a de a = r ndo se anula identicamente.

O documento — um bloco.

manifold M dim 4
bundle TM on M dim 4
chart sph on M coords (t, r, theta, phi)
metric g on sph bundle TM {
[t,t] = -exp(2*Phi(r)),
[r,r] = exp(2*Lambda(r)),
[theta, theta] = r*2,
[phi,phi] = r*2 * sin(theta)”2
b
tensor T : TM, TM on sph {
[t,t] = rho(r)*exp(-2*Phi(r)),
[r,r] = p(r)*exp(-2*Lambda(r)),
[theta, theta]l = p(r)/r~2,
[phi,phi] = p(r)/(r*2*sin(theta)”2)

}




A pergunta — o bloco seguinte, e a resposta.

eval T[a,b;b]

[*r] = (Phi'(r)*p(r) + Phi'(r)*rho(r) + p'(r))/exp(2*Lambda(r))
3 independent components identically zero

Uma sé componente nao nula, a de r, exatamente como o livro diz. Igualada a zero e
multiplicada por e/, d(p+p)+p =0, que éakEq. 10.27.

Esta conta encontrou o outro defeito, e o mais interessante. Antes desta rodada a saida
era (r*exp(-2A)*exp(2A)*®d'*p + ... + 2*¥exp(-2N\)*exp(2A)*p - 2*p)/(r*exp(2A)) : o par
métrica/inversa sobrevivia intacto em cada termo, e dois termos que deviam se cancelar
ficavam no resultado. Nao era resposta errada — era resposta numa forma que esconde o
cancelamento que a conta inteira existe para mostrar. Ver §6.2.

5. Capitulo 13 — cosmologia

A mesma maquina, aplicada ao universo inteiro: a métrica FRW, a equacao de Friedmann e a conservacao
da matéria.

5.1 O Einstein tridimensional da hipersuperficie | conrere
Schutz: §13.2, p.424, Eqg. 13.9

Ao construir a métrica cosmoldgica, Schutz precisa do tensor de Einstein da hipersuperficie
espacial dI? = e2NNdr2 + r2dQ?, e afirma gue ele se obtém do quadridimensional (Egs.
10.15-10.17) pondo @ = 0: Gy = —(1/r2)e? N1 — e™21), Ggg = —r €72\, Gy = sin6 Gop.

E uma afirmacdo sobre duas contas diferentes — uma em trés dimensdes, outra em

quatro — e vale conferir por isso.

O documento — um bloco.

manifold M dim 3
bundle TM on M dim 3
chart esp on M coords (r, theta, phi)
metric h on esp bundle TM {
[r,r] = exp(2*Lambda(r)),
[theta, theta] = r*2,
[phi,phi] = r*2 * sin(theta)”2

}

tensor G := einstein h

tensor S : TM*, TM* on esp {
[r,r] = -exp(2*Lambda(r))*(1 - exp(-2*Lambda(r)))/r"*2,
[theta,theta] = -r*exp(-2*Lambda(r))*Lambda'(r),
[phi,phi] = -sin(theta)”2*r*exp(-2*Lambda(r))*Lambda'(r)




A pergunta — o bloco seguinte, e a resposta.

einstein

eval G[a,b] - S[a,b]

G[r,r] = (1 - exp(2*Lambda(r)))/r"2

G[theta, theta] = -r*Lambda'(r)/exp(2*Lambda(r))
G[phi,phi] = -r*sin(theta)”2*Lambda'(r)/exp(2*Lambda(r))
3 independent components identically zero

9 independent components identically zero

Nove de nove nulas: a conta em trés dimensdes da o mesmo que a de quatrocom ® =0, e
o atalho do livro esté certo.

5.2 A componente temporal do Einstein em FRW | conFERE
Schutz: §13.3, p.439, Eq. 13.51

Gyt = 3(R/R)? + 3k/R2. E a Unica componente que a cosmologia precisa calcular: a isotropia
mata as mistas, e a identidade de Bianchi liga a espacial a esta.

O documento — um bloco.

manifold M dim 4
bundle TM on M dim 4
chart frw on M coords (t, r, theta, phi)
metric g on frw bundle TM {
[t,t] = -1,
[r,r] Scale(t)”2/(1 - k*r~2),
[theta,theta] = Scale(t)”2*r*2,
[phi,phi] = Scale(t)”2*r*2*sin(theta)”2
b
tensor G := einstein g
tensor S : TM*, TM* on frw { [t,t] = 3*diff(Scale(t), t)~2/Scale(t)”2 +
3*k/Scale(t)”2 }

A pergunta — o bloco seguinte, e a resposta.

eval G[a,b]

eval G[a,b] - S[a,b]

[t,t] (3*k + 3*Scale'(t)"2)/Scale(t)"2

[r,r] = (-k - 2*Scale(t)*Scale''(t) - Scale'(t)"2)/(1 - k*r"2)
[theta,theta] = -k*r*2 - 2*r”2*Scale(t)*Scale''(t) - r"2*Scale'(t)”"2
[phi,phi] = -k*r*2*sin(theta)”2 - 2*r"2*sin(theta)”2*Scale(t)*Scale''(t) -
r~2*sin(theta)”2*Scale' (t)"2

12 independent components identically zero




(3k + 3R2)/R? é a Eq. 13.51. Na segunda chamada, a componente [t,t] desaparece da
lista de ndao nulas — as trés que sobram sao as espaciais, que S nao declara.

As doze mistas nulas confirmam de passagem a outra afirmacao do livro, a de que a
isotropia garante Gy = 0.

5.3 A equacao de movimento da matéria | conFere
Schutz: §13.3, p.439, Eq. 13.47

De TH'., = 0 em FRW sai d(pR3)/dt = —p d(R?)/dt: a taxa de variacado da energia de um

elemento de fluido é o trabalho que ele faz ao se expandir. Schutz nota que, por isotropia,
as componentes espaciais se anulam identicamente.

O documento — um bloco.

manifold M dim 4
bundle TM on M dim 4
chart frw on M coords (t, r, theta, phi)
metric g on frw bundle TM {
[t,t] = -1,
[r,r] Scale(t)"2/(1 - k*r~2),
[theta, theta] = Scale(t)"2*r"2,
[phi,phi] = Scale(t)”2*r"2*sin(theta)”"2
}
tensor T : TM, T™ on frw {
[t,t] = rho(t),
[r,r] = p(t)*(1 - k*r*2)/Scale(t)"2,
[theta, theta]l = p(t)/(Scale(t)"2*r"2),
[phi,phi] = p(t)/(Scale(t)"2*r"2*sin(theta)”"2)
}

A pergunta — o bloco seguinte, e a resposta.

eval T[a,b;b]

[*t] = (Scale(t)*rho'(t) + 3*Scale'(t)*p(t) + 3*Scale'(t)*rho(t))/Scale(t)
3 independent components identically zero

Uma componente nao nula, a temporal, e as trés espaciais nulas — as duas metades da
afirmacé&o. Dividindo por Scale(t) : p + 3(R/R)(p + p) = 0. Multiplicando por R3: pR3 +
3pR2R + 3pR2R = 0, que € d(pR3)/dt = —p d(R3)/dt, a Eq. 13.47.

O caso da poeira (p = 0) da d(pR3)/dt = 0, e o da radiacao (p = p/3) da d(pR#)/dt = 0 — as
Eqgs. 13.48 e 13.50, que saem desta por substituicao.

6. O que nao funcionou, e o que foi feito

Duas das doze contas nao saiam. Nenhuma das duas por falta de recurso no programa: as
duas por defeito, e uma delas silencioso.



6.1 diff era um simbolo opaco

Escrever a Eq. 10.14 como o livro a escreve exige a derivada d/dr[r(1 - e2")] . O manual
documenta diff(f(r), r) na §9.10, e a recusa de sin'(x) diz, com todas as letras, "write
diff(sin(...), var)'

Escrever isso num documento produzia:
[ diff(sin(x), x) -> diff(sin(x),x)

Um simbolo opaco de dois argumentos chamado diff, jamais derivado. Quem seguisse a
mensagem de erro recebia uma resposta de aparéncia plausivel e silenciosamente errada
— exatamente a classe de falha que a lista de nomes reservados do projeto existe para
impedir no caso de tan.

Resolvido: diff(EXPR, VAR) agora aplica a derivada, ligada ao diff que o motor ja tinha.
O segundo argumento é um nome de variavel e sé isso — diff(f(x), x+1) é recusado
nomeando a grafia certa, em vez de tropecar num ) faltando.

diff(sin(x), x) ->  cos(x)

diff(f(x)"2, x) -> 2%f(x)*f' (x)

diff(f(x)* ( ), X) ->  f(x)*h'(x) + f'(x)*h(x)
diff(diff(f(x), x), x) ->  f''(x)

diff(f(x), y) -> 0

6.2 exp € a equacao TOV

O motor tratava exp(u) como atomo opaco, sem relacdo nenhuma com exp(-u) . Numa
métrica estatica isso é fatal de um jeito particular: g rr = €2 e ainversa traz e 2", e
toda divergéncia multiplica os dois. A TOV saia assim:

(r*exp(-2*Lambda(r))*exp(2*Lambda(r))*Phi' (r)*p(r) + ...
+ 2*exp(-2*Lambda(r))*exp(2*Lambda(r))*p(r) - 2*p(r))/(r*exp(2*Lambda(r)))

O par intacto em cada termo, e os dois Gltimos — que se cancelam — de pé no resultado.
Resposta certa, numa forma que esconde o cancelamento que a conta existe para mostrar.

A primeira tentativa, e por que ela era pior que o problema

A regra 6bvia é exp(a)-exp(b) = exp(a+b) , aplicada na multiplicacdo de mondmios. Ela fez
a TOV fechar — e quebrou o resto. Reprodutor minimo:

(exp(x) + exp(y))"2

2*exp (x)*exp(y)*exp(x + y)~4*exp(2*x + 2*y)”"3*exp(4*x + 4*y)"3
*exp (8*x + 8*y)"3*...*exp(1024*x + 1024*y)

O argumento dobrando até 1024, e com valor errado. A causa é estrutural e o comentario
gue a regra tinha sido deixada de fora ja a antecipava: exp(x), exp(y) , exp(x+y) €

exp(2x+2y) sao geradores algebricamente dependentes, e o0 anel de polindbmios sobre
0 qual o MDC trabalha supde geradores independentes. Cada rodada de reducao fundia o
gue a anterior havia criado. O tensor de Einstein da §10.2 saia errado por isso.

O que resolve

Sinal cano6nico no argumento: exp(-u) é internado como o mesmo gerador que
exp(u) , com expoente —1. Nao cria gerador nenhum — apenas deixa de criar um a mais
— e quem corta o par é o MDC comum gue ja existia.

exp(x)*exp(-x) -> 1

exp(x-y)*exp(y-x) -> 1

exp(x)"3*exp(-x)"2 ->  exp(x)

(exp(x)+exp(y))"2 ->  2*exp(x)*exp(y) + exp(x)™2 + exp(y)~2

E a TOV fecha:

[ [*r] = (Phi'(r)*p(r) + Phi'(r)*rho(r) + p'(r))/exp(2*Lambda(r))

6.3 O que continua em aberto

exp(2x)/exp(x) nao reduz. O sinal candnico junta u com —u; nao junta u com 2u. Fechar
isso significa extrair o conteudo inteiro do argumento — exp(2u) - exp(u)? —, 0 que é
uma decisao de representacao e nao um remendo: pde um céalculo de conteldo em toda
internacdo de exp, que é caminho quente, e muda a grafia de toda saida com exponencial
( exp(2*Lambda(r)) viraria exp(Lambda(r))~2, mais longe do que o livro escreve). Nenhuma
das doze contas precisa disso, e o limite esta fixado por um teste que falha se alguém o
fechar sem atualizar a decisao.

Duas outras contas do livro ficaram de fora por nao serem do tipo que este programa
responde. A §8.3 (campo fraco) pede expansao em série, e o ODEROM nao expande em
série. A §7.4 (quantidades conservadas) pede vetores de Killing, que nao existem como
consulta.

7. ldentidades e fatos estruturais — dez contas de outro
tipo

As doze anteriores comparavam um tensor calculado com uma férmula do livro. Estas dez sdao de outra
natureza: identidades que o Riemann satisfaz por construcdo, curvatura nula onde a conexao nao é nula,
matéria em espaco plano, e uma solucao exata sem simetria esférica.

7.1 Curvatura nula onde a conexao nao é nula | conFere
Schutz: §6.5, p.159, logo abaixo da Eq. 6.71; Exercicio 6.30, p.167

R“B“\, = 0 < variedade plana. Schutz sugere entre parénteses: "try showing that this is

true in polar coordinates for the Euclidean plane", e o Exercicio 6.30 pede o mesmo para o
cilindro.

E a conta que separa "a conexao nao é nula" de "o espaco é curvo". Em polares os
trés simbolos de Christoffel da §2.1 sdo nao nulos, e mesmo assim o Riemann se anula: os
termos quadraticos I'T cancelam exatamente as derivadas ar.

O documento — um bloco.

manifold M dim 2

bundle TM on M dim 2

chart pol on M coords (r, theta)

metric g on pol bundle ™M { [r,r] = 1, [theta,theta]l = r*2 }
tensor R := riemann g

# e o cilindro, num documento separado:

manifold M dim 2

bundle TM on M dim 2

chart cil on M coords (z, phi)

metric g on cil bundle ™ { [z,z] = 1, [phi,phi] = a"2 }
tensor R := riemann g

A pergunta — o bloco seguinte, e a resposta.

eval R[a,b,c,d]

eval R[a,b,c,d]

16 independent components identically zero

16 independent components identically zero

O controle esta no teste, e é a esfera: a MESMA maquina, contra a“2*sin(theta)~2, nao da
zero (§3.1). Sem ele, um riemann que devolvesse zero sempre passaria nos dois.

E o teste mais afiado deste documento para uma classe especifica de defeito: apagar os
termos quadraticos ' do Riemann deixa o cilindro passando — ele é plano por
coeficientes constantes — e derruba o plano em polares na hora. Foi conferido assim,
sabotando.

7.2 A trés-esfera tem curvatura escalar constante CONFERE
Schutz: Exercicio 6.33, p.167

A trés-esfera de raio a, em coordenadas hiperesféricas: Oyx = @°, ggg = @% SiN%X, gpe = @2
sin?y sin20. A curvatura escalar é 6/a2.

Constante, e é isso que faz dela um espaco homogéneo e isotrépico — o caso k = +1 da
cosmologia do capitulo 13.

O documento — um bloco.



manifold M dim 3
bundle TM on M dim 3
chart s3 on M coords (chi, theta, phi)
metric g on s3 bundle T™M {
[chi,chi] = a"2,

[theta, theta] = a”2*sin(chi)"2,
[phi,phi] = a”2*sin(chi)”2*sin(theta)”"2
}

tensor Ric := ricci g

A pergunta — o bloco seqguinte, e a resposta.

eval Ricl[a,a]
[ 6/a"2

Sem y nem 6 na resposta, que é metade da afirmacao: a curvatura é a mesma em todo
ponto.

7.3 A divergéncia pela raiz do determinante | conFere
Schutz: §6.3, p.151, Eqs. 6.40 e 6.42

Va;cx = (1/V]g]) (V]g| Va),cx-

Ela sai de F“ua = (\/—g),u/\/—g (Eqg. 6.40): o traco do Christoffel é a derivada logaritmica do
determinante da métrica. E a forma que se usa na pratica, porque dispensa calcular T.

O documento — um bloco.

manifold M dim 4
bundle TM on M dim 4
chart schw on M coords (t, r, theta, phi)
metric g on schw bundle T™M {
[t,t] = -(1 - 2*Mass/r),
[r,r] = 1/(1 - 2*Mass/r),
[theta, theta] = r*2,
[phi,phi] = r*2 * sin(theta)”2

}

tensor V. : TM on schw { [t] = A(t,r), [r] = B(t,r), [theta] = C(theta), [phi] =
D(phi) }
raiz := sqrt(r"4*sin(theta)”2)
tensor S : TM on schw {
[t] = (diff(raiz*A(t,r), t) + diff(raiz*B(t,r), r)
+ diff(raiz*C(theta), theta) + diff(raiz*D(phi), phi))/raiz

A pergunta — o bloco seqguinte, e a resposta.



eval V[a;al

eval S[al

(r*sin(theta)*A t(t,r) + r*sin(theta)*B r(t,r) + r*sin(theta)*C'(theta)
+ r*sin(theta)*D' (phi) + r*cos(theta)*C(theta) +
2*sin(theta)*B(t,r))/(r*sin(theta))

[~t] = (r*sin(theta)*A t(t,r) + r*sin(theta)*B r(t,r) + r*sin(theta)*C'(theta)
+ r*sin(theta)*D' (phi) + r*cos(theta)*C(theta) +
2*sin(theta)*B(t,r))/(r*sin(theta))

Idénticas, caractere por caractere — e calculadas por caminhos gque nao se tocam: a
primeira pelos simbolos de Christoffel, a segunda por sqrt e diff.Em Schwarzschild —g
= r4sinZ0.

Os dois ultimos termos sao os que a derivada parcial sozinha nao da:
r*cos(theta)*C(theta) e 2*sin(theta)*B(t,r), que vém inteiros da raiz do determinante.

7.4 O fluido perfeito, da forma invariante a diagonal | conFere

Schutz: §4.7, p.100, Eqgs. 4.36-4.38

A forma invariante é T®® = (p+p)U%UP + p n®*P (Eq. 4.37), e no referencial de repouso ela
tem de dar T = diag(p, p, p, p) (Eq. 4.36).

Schutz verifica isso a mao, componente a componente: "ifa = B = 0 then U° = 1, noo =

—1, and T®B = (p + p) —p = p... by trying all possible a and B you can verify".

O documento — um bloco.

manifold M dim 4
bundle TM on M dim 4
chart mink on M coords (t, x, y, 2)

metric g on mink bundle ™M { [t,t] = -1, [x,x] =1, [y,yl =1, [z,z] =1
tensor U : TM on mink { [t] =1 }
tensor T~{\mu\nu} := (rho + p) U {\mu} U~{\nu} + (p) g~ {\mu\nu}

A pergunta — o bloco seguinte, e a resposta.

eval T[a,b]

eval T[a,al
eval T[a,b] U[a] U[b]




[~t,~t] = rho
[*x,”x] = p
[*y,%y]l = p
[*z,7z] = p

12 independent components identically zero
3*p - rho

rho

As dezesseis componentes, e as doze fora da diagonal nulas — o "by trying all possible a
and B" feito de uma vez. O traco 3p — p e a densidade gque o préprio fluido mede, p, saem
da mesma declaracao.

Esta conta nao podia ser escrita até a rodada passada. Um monomio sé aceitava
coeficiente racional, e (rho + p) é simbdlico. O coeficiente escalar entre parénteses fecha
isso; a §8 conta o que era o problema.

7.5 As simetrias do tensor de Riemann CONFERE
Schutz: §6.5, p.158, Egs. 6.69 € 6.70
R“BUV = _RBQUV = _RGBVU = RHVQB (6.69), e RGBUV + RQVBIJ + RGUVB =0 (6.70).

Sao elas que reduzem as 256 componentes de um tensor de rank 4 em quatro dimensdes
as 20 independentes.

O documento — um bloco.

manifold M dim 4
bundle TM on M dim 4
chart schw on M coords (t, r, theta, phi)
metric g on schw bundle TM {
[t,t] = -(1 - 2*Mass/r),

[r,r] = 1/(1 - 2*Mass/r),

[theta, theta] = r*2,

[phi,phi] = r*2 * sin(theta)”2
b

tensor R := riemann g

A pergunta — o bloco seguinte, e a resposta.

eval R[a,b,c,d] R[b,a,c,d]
eval R[a,b,c,d] R[a,b,d,c]

eval R[a,b,c,d] R[c,d,a,b]
eval R[a,b,c,d] R[a,d,b,c] + R[a,c,d,b]




256 independent components identically zero
256 independent components identically zero
256 independent components identically zero
256 independent components identically zero

Schwarzschild tem 24 células ndao nulas no Riemann, entao nenhuma das quatro
identidades fecha por vacuidade — o teste confere isso antes.

7.6 As identidades de Bianchi, na forma diferencial CONFERE
Schutz: §6.6, p.162, Eq. 6.90

RaBuwA + RaBAuN + RanAm1= 0.

E a forma de onde tudo o mais vem: contraida uma vez da a Eq. 6.93, contraida duas
vezes da V-G = 0 (§3.3), e é por causa dela que as equacoes de Einstein sao compativeis
com a conservacgao da energia.

O documento — um bloco.

manifold M dim 4
bundle TM on M dim 4
chart schw on M coords (t, r, theta, phi)
metric g on schw bundle T™M {
[t,t] -(1 - 2*Mass/r),

[r,r] 1/(1 - 2*Mass/r),

[theta, theta] = r"2,

[phi,phi] = r*2 * sin(theta)”2
}

tensor R := riemann g

A pergunta — o bloco seguinte, e a resposta.

eval R[a,b,c,d;e] + R[a,b,d,e;c] + R[a,b,e,c;d]

(491.984 ms)
1024 independent components identically zero

Mil e vinte e quatro células, cada uma uma derivada covariante de um tensor de rank 4 —
a conta mais pesada deste documento, e meio segundo.

7.7 O traco do Einstein espacial é constante | conFerE
Schutz: §13.2, p.425, Eq. 13.11

A condicao de homogeneidade da cosmologia: o traco do tensor de Einstein tridimensional
tem de ser uma constante k. Para g, = 1/(1 — k r?) ele vale —3k.

E de onde Schutz tira a definicdo k := —«k/3, e com ela a métrica espacial da Eq. 13.13.

O documento — um bloco.



manifold M dim 3
bundle TM on M dim 3
chart esp on M coords (r, theta, phi)
metric h on esp bundle TM {
[r,r] = 1/(1 - k*r*2),

[theta, theta] = r*2,
[phi,phi] = r*2 * sin(theta)”2
}

tensor G := einstein h

A pergunta — o bloco seqguinte, e a resposta.

eval G[a,a]
[ 3

Sem r na resposta, que é a metade da afirmacao que importa: a curvatura é a mesma em
todo ponto, e é isso que "homogéneo" quer dizer.

7.8 A constante cosmoldgica € um fluido de pressao negativa
CONFERE
Schutz: §13.3, p.440, Eqs. 13.53-13.54

TA%® = —(A/8m) g% tem pp = A/8TL e pp = —pp.

A pressdo negativa é o ponto inteiro — é ela que acelera a expansao.

O documento — um bloco.

manifold M dim 4
bundle TM on M dim 4
chart frw on M coords (t, r, theta, phi)
metric g on frw bundle TM {
[t,t] = -1,
[r,r] = Scale(t)”2/(1 - k*r*2),
[theta,theta] = Scale(t)”2*r"2,
[phi,phi] = Scale(t)”2*r"2*sin(theta)”"2
}
tensor U : TM on frw { [t] =1 }
tensor TL : TM, TM on frw {
[t,t] Lam/ (8*pi),
[r,r] -Lam*(1 - k*r"2)/(8*pi*Scale(t)"2),
[theta,theta] = -Lam/(8*pi*Scale(t)”"2*r"2),
[phi,phi] = -Lam/(8*pi*Scale(t)”2*r"~2*sin(theta)”"2)

}

A pergunta — o bloco seguinte, e a resposta.



eval TL[a,b] U[a] U[b]
eval TL[a,a]

eval TL[a,b;b]

Lam/ (8*pi)
-Lam/ (2*pi)

4 independent components identically zero

A primeira é pp = A/8m. A segunda € o traco 3p — p, e —A/2n = —4-(\/8m) s6 fecha com p =
—p. A terceira é a conservacao, que vale de graca porque Vg = 0.

7.9 A métrica de Kerr é vacuo | CcONFERE
Schutz: §11.3, p.328, Eq. 11.71

O buraco negro em rotacao, em coordenadas de Boyer-Lindquist, com A =r2 — 2Mr + a2 e
P2 = r2 + a2cos26. E solucdo de vacuo: Rqg = 0.

E a Unica solucdo exata deste documento que ndo é esfericamente simétrica: tem termo
cruzado gy, — 0 arrasto de referenciais — e um denominador que depende de duas

coordenadas.

O documento — um bloco.

Sigma := r"2 + a”2*cos(theta)”2
Delta := r*2 - 2*M*r + a2

manifold M dim 4
bundle TM on M dim 4
chart bl on M coords (t, r, theta, phi)
metric g on bl bundle TM {
[t,t] = -(1 - 2*M*r/Sigma),
[t,phi] = -2*M*a*r*sin(theta)”~2/Sigma,
[r,r] = Sigma/Delta,
[theta, theta] = Sigma,
[phi,phi] = (r*2 + a®2 + 2*M*a"2*r*sin(theta)”2/Sigma) * sin(theta)”2
}

tensor Ric := ricci g

A pergunta — o bloco seqguinte, e a resposta.

eval Ric[a,b]

motor: localizado
16 independent components identically zero




Menos de um segundo. A linha motor: localizado diz qual dos dois motores rodou: o
denominador bivariado de Kerr é justamente o caso que a forma racional geral nao fecha,
e para o qual existe o motor localizado.

O documento acima é examples/kerr.od , 0 mesmo que o resto da suite usa — nao uma
cbdpia que possa divergir dele.

7.10 A divergéncia de um tensor antissimétrico | conrere
Schutz: Exercicio 6.34(c), p.166

Para FH¥ antissimétrico, F*V., = (V—g F¥") ,/V—g.

E a divergéncia de Maxwell em espaco curvo, e a razdo de as equacdes do campo
eletromagnético nao precisarem de simbolo de Christoffel nenhum. A identidade sé vale
por causa da antissimetria: os termos de Christoffel que sobrariam sao simétricos nos dois
indices e morrem contra ela.

O documento — um bloco.

manifold M dim 4
bundle TM on M dim 4
chart schw on M coords (t, r, theta, phi)
metric g on schw bundle TM {
[t,t] = -(1 - 2*Mass/r),
[r,r] 1/(1 - 2*Mass/r),
[theta, theta] = r*2,
[phi,phi] = r*2 * sin(theta)”2

b
tensor F : TM, TM on schw {

[t,r] = A(t,r,theta,phi), [r,t] = -A(t,r,theta,phi),
[t,theta]
[t,phi] =
[r,theta]

= B(t, r,theta,phi), [theta,t] = -B(t,r,theta,phi),
C(t,r,theta,phi), [phi,t] = -C(t,r,theta,phi),
= D(t, r,theta,phi), [theta,r] -D(t, r,theta,phi),

[r,phi] E(t,r,theta,phi), [phi,r] = -E(t,r,theta,phi),

[theta,phi] = H(t,r,theta,phi), [phi,theta] = -H(t,r,theta,phi)
b
raiz := sqrt(r™4*sin(theta)”2)
tensor S : TM on schw {

[t] = (diff(raiz*A(..), r) + diff(raiz*B(..), theta) + diff(raiz*C(..),
phi))/raiz,

[r] = (diff(-raiz*A(..), t) + diff(raiz*D(..), theta) + diff(raiz*E(..)
phi))/raiz,

[theta] = (diff(-raiz*B(..), t) + diff(-raiz*D(..), r) + diff(raiz*H(..),
phi))/raiz,

[phi] = (diff(-raiz*C(..), t) + diff(-raiz*E(..), r) + diff(-raiz*H(..),
theta))/raiz
b

A pergunta — o bloco seguinte, e a resposta.

eval F[a,b;b] - S[al



[ 4 independent components identically zero

As seis componentes independentes de F sao funcdes indeterminadas das quatro
coordenadas — nada se anula por acaso, e a divergéncia sozinha ndo é nula (o teste
confere isso antes). Acima (..) abrevia (t,r,theta,phi), que no arquivo real vai escrito
por extenso.

8. O laco que fechou

Uma das dez nao p6de ser escrita como o livro a escreve, e o motivo valia mais do que a
conta. A Eq. 4.37, T*® = (p+p)U%UP + p n®B, n&o tinha grafia neste programa: um
monomio sé aceitava coeficiente racional. Com um numero funcionava; com um
simbolo, nao.

tensor T~{\mu\nu} :
tensor T~{\mu\nu} :

3 U{\mu} U~{\nu} -> ok
rho U~{\mu} U~{\nu} -> unknown tensor head “rho’

E a saida que o programa oferecia ndo existia. Declarar o escalar como tensor de rank 0
era recusado, e a recusa mandava usar um apelido; o apelido era recusado onde a
mensagem mandava usa-lo. Trés mensagens, nenhuma errada por si, e um laco
fechado.

O que foi feito

O coeficiente escalar passou a existir, entre parénteses:

eval (rho + p) Ula]l U[b] + (p) gla,b]
tensor T~{\mu\nu} := (rho + p) U~{\mu} U~{\nu} + (p) g”~{\mu\nu}

Ele aceita qualquer expressao escalar — apelido, funcao conhecida, fracao —, multiplica o
termo inteiro (contracdes incluidas), e é aplicado depois da avaliacao: (rho) U[a] U[a] é
—p, hao p, porque U-U = —1.

Onde ele NAO vale, e por qué

O escalar viaja ao lado do mon6mio, ndo dentro dele. O coeficiente de um monémio é um
racional de maquina de propdsito: ele entra no Eq/Hash e na ordenacao de Butler-
Portugal, e troca-lo por expressao simbdlica seria mudar o tipo central do nucleo, com
consequéncias em simplify, solve e na canonicalizacao.

Entdao quem avalia componentes multiplica a grade pelo escalar, e o lado abstrato recusa
— dizendo onde a grafia vale, que é o que faltava nas trés mensagens antigas:

simplify (rho + p) T[a,b]

error: um coeficiente escalar entre parenteses -- “(rho + p) U {\mu} U {\nu} " --
so' vale onde ha COMPONENTES para multiplicar: “eval® e o lado direito de
“tensor NOME~{...} := ... . No lado abstrato ( simplify , “solve’, “canon’)

um simbolo solto nao tem significado sem um head de rank 0 que o declare
(“head NOME : on M")

As outras duas mensagens do laco passaram a apontar para a saida que agora existe. Um
nome solto em posicao de fator sugere os parénteses; a recusa do rank 0 diz as duas
coisas — como declarar o escalar e como usa-lo:

tensor T*{\mu\nu} := rho U~{\mu} U~{\nu}
error: unknown tensor head ‘rho” -- se “rho’ era para ser um coeficiente
ESCALAR, ele vai entre parenteses: " (rho) T~{\mu\nu}"

tensor E : on mink { [] = rho + p }

error: "E° teria rank O (um escalar). Um escalar com valor se declara como
apelido ('E := EXPR"), e entra numa expressao tensorial como
COEFICIENTE, entre parenteses: " (E) T {\mu\nu}"

A dica dos parénteses sé aparece quando o fator nao tem indices. Quem escreveu
Ql[a,b] quis um tensor, e sugerir um coeficiente ali seria conselho errado — tem teste
préprio dizendo isso.

